Numerical, Graphical and Symbolic Analysis of Bernoulli Equations[1]

Douglas B. Meade[2]

Introduction

The problem discussed in this paper is an extension of a homework problem in a sophomore-level course in differential equations. The exercises at the end of the section on the improved Euler's method for solving first-order initial value problems include ([1], p. 109)

> **19. Logistic Model.** In Section 3.2 we discussed the logistic equation
>
> $$\frac{dp}{dt} = ap - bp^2, \qquad p(0) = p_0$$
>
> and its use in modeling population growth. A more general model might involve the equation
>
> $$\frac{dp}{dt} = ap - bp^r, \qquad p(0) = p_0, \tag{1}$$
>
> where $r > 1$. To see the effect of changing the parameter r in (1), take $a = 3$, $b = 1$, and $p_0 = 1$. Now use the improved Euler's method with $h = 0.25$ to approximate the solution to (1) on the interval $0 \le t \le 5$ for $r = 1.5$, 2, and 3.

The question, as posed, is routine. However, in the course of implementing a solution in Maple, it quickly becomes apparent that this problem is a perfect vehicle to display the complementary roles of numerical, graphical, and analytical techniques in the solution of differential equations. In this paper I will discuss how I use Maple to assist my students in the analysis of this problem. In particular, we will see how Maple can be coaxed into providing information it seems unwilling to provide.

The Original Problem

Numerical solutions to an initial value problem can be obtained by Maple's `dsolve` command (see `?dsolve[numeric]` for more information). However, the improved Euler's method is not one of the built-in numerical methods known to `dsolve`. A pseudo-code implementation of the improved Euler's method for the approximate solution of the first-order initial value problem

$$\frac{dy}{dx} = f(x, y), \qquad y(x_0) = y_0$$

[1]This work partially supported by the National Science Foundation under grant DMS-9404488.

[2]Department of Mathematics, University of South Carolina, Columbia, SC 29208, USA,
E-mail: `meade@math.scarolina.edu`, WWW URL: `http://www.math.scarolina.edu/~meade/`

is provided in the text [1, p. 105]. This algorithm is easily translated into the Maple programming language.

```
> IEM := proc( f:procedure, ic:[numeric,numeric], h:numeric, N:nonnegint )
>   local F, G, n, new_pt, pts, x, y;
>   x[0] := ic[1];
>   y[0] := ic[2];
>   pts := ic;
>   for n from 0 to N-1 do
>     F := f(x[n],y[n]);
>     G := f(x[n]+h,y[n]+h*F);
>     x[n+1] := x[n] + h;
>     y[n+1] := y[n] + h/2*(F+G);
>     new_pt := [ x[n+1], y[n+1] ];
>     pts := pts, new_pt;
>   od:
>   RETURN( [pts] );
> end:
```

As this course is most students' first exposure to Maple, I provide a skeleton of the procedure and have the students work in groups to fill in the missing details. (The type checking seems to be understandable, and beneficial, for students.) We are now ready to begin to use this procedure to answer the original question.

```
> f := (t,p) -> a*p-b*p^r;
```

$$f := (t, p) \to a\,p - b\,p^r$$

```
> a := 3: b := 1:
> r := 3/2;
```

$$r := \frac{3}{2}$$

```
> s1 := IEM( f, [0, 1], 0.25, 20 );
```

$$\begin{aligned}s1 := [&[0,1], [.25, 1.582860337], [.50, 2.351441074], [.75, 3.267497751],\\ &[1.00, 4.253155546], [1.25, 5.216751357], [1.50, 6.083401955], [1.75, 6.811626143],\\ &[2.00, 7.392146417], [2.25, 7.837089763], [2.50, 8.168506666], [2.75, 8.410361872],\\ &[3.00, 8.584317467], [3.25, 8.708165353], [3.50, 8.795710426], [3.75, 8.857285190],\\ &[4.00, 8.900442869], [4.25, 8.930618540], [4.50, 8.951681525], [4.75, 8.966366415],\\ &[5.00, 8.976596163]]\end{aligned}$$

```
> r := 2;
```

$$r := 2$$

```
> s2 := IEM( f, [0, 1], 0.25, 20 );
```

$$\begin{aligned}s2 := [&[0,1], [.25, 1.531250000], [.50, 2.049596779], [.75, 2.440026760],\\ &[1.00, 2.686753607], [1.25, 2.829199430], [1.50, 2.908038129], [1.75, 2.950802194],\\ &[2.00, 2.973767211], [2.25, 2.986036664], [2.50, 2.992574319], [2.75, 2.996052946],\\ &[3.00, 2.997902518], [3.25, 2.998885541], [3.50, 2.999407895], [3.75, 2.999685431],\\ &[4.00, 2.999832881], [4.25, 2.999911217], [4.50, 2.999952834], [4.75, 2.999974943],\\ &[5.00, 2.999986688]]\end{aligned}$$

```
> r := 3;
```

$$r := 3$$

```
> s3 := IEM( f, [0, 1], 0.25, 20 );
```

$s3 := [[0,1],[.25,1.390625000],[.50,1.553471728],[.75,1.628846948],$
$[1.00,1.669991926],[1.25,1.694055606],[1.50,1.708577858],[1.75,1.717479239],$
$[2.00,1.722980373],[2.25,1.726395709],[2.50,1.728521701],[2.75,1.729847172],$
$[3.00,1.730674332],[3.25,1.731190820],[3.50,1.731513436],[3.75,1.731714998],$
$[4.00,1.731840945],[4.25,1.731919650],[4.50,1.731968837],[4.75,1.731999577],$
$[5.00,1.732018789]]$

A moment's reflection on these numerical data suggests that the limiting population (the capacity of the environment) decreases as the exponent r increases. This point is further reinforced by a graphical presentation of the results.

```
> plot( {s1,s2,s3}, title=`Figure 1` );
```



Many students are able to recognize that the limiting populations are $p = 9$, $p = 3$, and $p = \sqrt{3}$ for the three values of r. At this point the original questions have been answered.

Extensions of the Problem

It would be a shame not to consider a number of natural extensions to this problem. This problem is particularly attractive because students are able to formulate conjectures about properties of the solution and verify their conjectures using their current knowledge.

Most students immediately notice that all three limiting populations are powers of 3; the precise connection between r and the power is not completely trivial. With a little guidance, the class decides to narrow their attention to the following issues:

- Why are the limiting populations all powers of 3?
- What is the limiting population for a general power r?
- Does the limiting population depend on the initial population, p_0, and the parameters a and b?

A conjecture for the answer to the third question can be obtained from a plot of the direction field for the differential equation. The built-in `DEtools` package provides a number of useful procedures for the visualization of solutions to a differential equation.

```
> with( DEtools ):
```

```
> r := 3/2:
> DEplot( f(t,p), [t,p], t=0..5, p=0..15, arrows=THIN,
>         title=`Figure 2a: Direction field for r=1.5` );
```




```
> r := 2:
> DEplot( f(t,p), [t,p], t=0..5, p=0..15, arrows=THIN,
>         title=`Figure 2b: Direction field for r=2` );
```



```
> r := 3:
> DEplot( f(t,p), [t,p], t=0..5, p=0..15, arrows=THIN,
>         title=`Figure 2c: Direction field for r=3` );
```



These plots (together with a little analysis) illustrate that for each of these values of r there is a unique positive steady-state solution that is asymptotically stable for all initial conditions $p_0 > 0$. If $0 < r < 1$ there is still a unique positive steady-state solution, but, as Figure 2d illustrates, this solution is unstable.

```
> r := 1/2:
> DEplot( f(t,p), [t,p], t=0..5, p=0..0.5, arrows=THIN,
>         title=`Figure 2d: Direction field for r=1/2` );
```



The Search For A Solution

A solution (either implicit or explicit) to the initial value problem will both reinforce the previous observations and provide a means to answer the remaining questions. To find an exact solution to this Bernoulli equation, the `dsolve` command seems appropriate. However, Maple is unable to provide the solution to the IVP, even when the known assumptions on the parameters are specified.[3]

```
> a := 'a': b := 'b': r := 'r': p0 := 'p0':
> ode := diff( p(t), t ) = f( t, p(t) );
```

$$ode := \frac{\partial}{\partial t}\,\mathrm{p}(t) = a\,\mathrm{p}(t) - b\,\mathrm{p}(t)^r$$

```
> ic := p(0)=p0;
```

$$ic := \mathrm{p}(0) = p0$$

```
> #infolevel[all] := 5:
> dsolve( {ode, ic}, p(t) );
> assume( a>0, b>0, r>1, p0>0 );
> dsolve( {ode, ic}, p(t) );
> a := 'a': b := 'b': r := 'r': p0 := 'p0':
> infolevel[all] := 0:
```

All is not lost; `dsolve` is able to find the general solution to the differential equation.

```
> GSOLN := dsolve( ode, p(t) );
```

$$GSOLN := -\frac{r\ln(\mathrm{p}(t))}{a\,(r-1)} + \frac{\ln\left(-a\,\mathrm{p}(t) + b\,\mathrm{p}(t)^r\right)}{a\,(r-1)} + t = \_C1$$

It is then possible to manually instruct Maple to solve for the constant of integration

```
> Cvalue := solve( subs( {t=0,p(t)=p0}, GSOLN ), { _C1 } );
```

$$Cvalue := \left\{\_C1 = -\frac{r\ln(p0)}{a\,(r-1)} + \frac{\ln(-a\,p0 + b\,p0^r)}{a\,(r-1)}\right\}$$

An implicit solution to the IVP is then found to be

```
> impSOLN := subs( Cvalue, GSOLN );
```

$$impSOLN := -\frac{r\ln(\mathrm{p}(t))}{a\,(r-1)} + \frac{\ln\left(-a\,\mathrm{p}(t) + b\,\mathrm{p}(t)^r\right)}{a\,(r-1)} + t = -\frac{r\ln(p0)}{a\,(r-1)} + \frac{\ln(-a\,p0 + b\,p0^r)}{a\,(r-1)}$$

This implicit solution appears to be of limited utility. The general behavior of solutions is not readily apparent, and Maple is unable to directly solve this equation for $p(t)$. We can, however, manipulate this expression into a more useful form.

The first step towards isolating $p(t)$ is to subtract t to both sides of the implicit solution.

```
> impSOLN - ( t=t );
```

$$-\frac{r\ln(\mathrm{p}(t))}{a\,(r-1)} + \frac{\ln\left(-a\,\mathrm{p}(t) + b\,\mathrm{p}(t)^r\right)}{a\,(r-1)} = -\frac{r\ln(p0)}{a\,(r-1)} + \frac{\ln(-a\,p0 + b\,p0^r)}{a\,(r-1)} - t$$

Next, combine all the logarithms, exponentiate both sides of the equation, and simplify the

[3] Uncomment the `infolevel` command to see all the decisions Maple makes in its attempt to find the requested solution.

expression when possible. (Note, however, that `simplify` does not help as much as `combine`.)

```
> combine( ", ln, anything );
```

$$\ln\left(\mathrm{p}(t)^{\left(-\frac{r}{a(r-1)}\right)}\left(-a\,\mathrm{p}(t)+b\,\mathrm{p}(t)^r\right)^{\left(\frac{1}{a(r-1)}\right)}\right)=-t+\ln\left(p0^{\left(-\frac{r}{a(r-1)}\right)}\left(-a\,p0+b\,p0^r\right)^{\left(\frac{1}{a(r-1)}\right)}\right)$$

```
> map( exp, " );
```

$$\mathrm{p}(t)^{\left(-\frac{r}{a(r-1)}\right)}\left(-a\,\mathrm{p}(t)+b\,\mathrm{p}(t)^r\right)^{\left(\frac{1}{a(r-1)}\right)}=\mathrm{e}^{\left(-t+\ln\left(p0^{\left(-\frac{r}{a(r-1)}\right)}\left(-a\,p0+b\,p0^r\right)^{\left(\frac{1}{a(r-1)}\right)}\right)\right)}$$

```
> combine( ", exp );
```

$$\mathrm{p}(t)^{\left(-\frac{r}{a(r-1)}\right)}\left(-a\,\mathrm{p}(t)+b\,\mathrm{p}(t)^r\right)^{\left(\frac{1}{a(r-1)}\right)}=p0^{\left(-\frac{r}{a(r-1)}\right)}\left(-a\,p0+b\,p0^r\right)^{\left(\frac{1}{a(r-1)}\right)}\mathrm{e}^{(-t)}$$

It is now beginning to look as though it should be possible to find an explicit solution. Maple is still unable to solve for $p(t)$; the basic reason is the simplification rules for non-integral exponents. Regardless, we can continue to step our way towards an explicit solution by raising both sides of the equation to a common (positive) power and simplifying.

```
> expand( lhs(")^(a*(r-1)) = rhs(")^(a*(r-1)) );
```

$$-\frac{a\,\mathrm{p}(t)}{\mathrm{p}(t)^r}+b=-\frac{(\mathrm{e}^t)^a\,a\,p0}{p0^r\,(\mathrm{e}^t)^{(r\,a)}}+\frac{(\mathrm{e}^t)^a\,b}{(\mathrm{e}^t)^{(r\,a)}}$$

```
> combine( ", power );
```

$$-\mathrm{p}(t)^{(-r+1)}\,a+b=-a\,p0^{(-r+1)}\,\mathrm{e}^{(-t\,a\,r+t\,a)}+b\,\mathrm{e}^{(-t\,a\,r+t\,a)}$$

This, finally, is simple enough that Maple can find an explicit solution to the ODE.

```
> SOLN := solve( ", p(t) );
```

$$SOLN:=\left(\frac{b+a\,p0^{(-r+1)}\,\mathrm{e}^{(-t\,a\,r+t\,a)}-b\,\mathrm{e}^{(-t\,a\,r+t\,a)}}{a}\right)^{\left(\frac{1}{-r+1}\right)}$$

A Second Approach

Prior to concluding this section on the search for explicit solutions to the Bernoulli equation it should be noted that the explicit solution can also be obtained by the classical technique involving the substitution $v=p^{1-r}$. Here is how that process could be implemented in Maple.

- STEP 1: Divide the ODE by p^r

```
> ode1 := ode/p(t)^r;
```

$$ode1:=\frac{\frac{\partial}{\partial t}\mathrm{p}(t)}{\mathrm{p}(t)^r}=\frac{a\,\mathrm{p}(t)-b\,\mathrm{p}(t)^r}{\mathrm{p}(t)^r}$$

- STEP 2: Make a substitution for the dependent variable.

```
> subs( p(t)=v(t)^(1/(1-r)), ode1 );
```

$$\frac{\frac{\partial}{\partial t}\mathrm{v}(t)^{\left(\frac{1}{-r+1}\right)}}{\left(\mathrm{v}(t)^{\left(\frac{1}{-r+1}\right)}\right)^r}=\frac{a\,\mathrm{v}(t)^{\left(\frac{1}{-r+1}\right)}-b\left(\mathrm{v}(t)^{\left(\frac{1}{-r+1}\right)}\right)^r}{\left(\mathrm{v}(t)^{\left(\frac{1}{-r+1}\right)}\right)^r}$$

```
> ode2 := simplify( combine( ", power ) );
```

$$ode2 := -\frac{\frac{\partial}{\partial t}\,\mathrm{v}(t)}{r-1} = \mathrm{v}(t)\,a - b$$

- STEP 3: Solve the linear ODE, with initial condition, using `dsolve`.

```
> dsolve( { ode2, v(0)=p0^(1-r) }, v(t) );
```

$$\mathrm{v}(t) = \frac{b}{a} + \left(-\frac{b}{a} + p0^{(-r+1)}\right) \mathrm{e}^{(-t\,a\,(r-1))}$$

- STEP 4: Invert the substitution and solve for p.

```
> subs( v(t)=p(t)^(1-r), " );
```

$$\mathrm{p}(t)^{(-r+1)} = \frac{b}{a} + \left(-\frac{b}{a} + p0^{(-r+1)}\right) \mathrm{e}^{(-t\,a\,(r-1))}$$

```
> solve( ", p(t) );
```

$$\left(-\frac{-b + \mathrm{e}^{(-t\,a\,(r-1))}\,b - \mathrm{e}^{(-t\,a\,(r-1))}\,p0^{(-r+1)}\,a}{a}\right)^{\left(\frac{1}{-r+1}\right)}$$

```
> SOLN2 := combine( ", exp );
```

$$SOLN2 := \left(\frac{b - \mathrm{e}^{(-t\,a\,(r-1))}\,b + \mathrm{e}^{(-t\,a\,(r-1))}\,p0^{(-r+1)}\,a}{a}\right)^{\left(\frac{1}{-r+1}\right)}$$

This solution is easily seen to be equivalent to the first solution.

```
> evalb( simplify(SOLN=SOLN2) );
```

$$true$$

The Final Analysis

The questions concerning the limiting population require some information about the behavior of the solution for large time.

```
> Plimit := map( Limit, p(t)=SOLN, t=infinity );
```

$$Plimit := \lim_{t\to\infty} \mathrm{p}(t) = \lim_{t\to\infty} \left(\frac{b + a\,p0^{(-r+1)}\,\mathrm{e}^{(-t\,a\,r+t\,a)} - b\,\mathrm{e}^{(-t\,a\,r+t\,a)}}{a}\right)^{\left(\frac{1}{-r+1}\right)}$$

```
> value( Plimit );
```

$$\lim_{t\to\infty} \mathrm{p}(t) = \lim_{t\to\infty} \left(\frac{b + a\,p0^{(-r+1)}\,\mathrm{e}^{(-t\,a\,r+t\,a)} - b\,\mathrm{e}^{(-t\,a\,r+t\,a)}}{a}\right)^{\left(\frac{1}{-r+1}\right)}$$

Maple is unable to evaluate this limit because it does not exist for all values of the parameters a, b, p_0, and r. (Remember that Maple considers all names to be complex-valued until otherwise indicated.)

```
> assume( a>0, b>0, p0>0, r>1 );
> value( Plimit );
```

$$\lim_{t\to\infty} \mathrm{p}(\,t\,) = \frac{e^{\left(\frac{\ln(\,a\tilde{}\,)}{r\tilde{}\,-1}\right)}}{b\tilde{}^{\left(\frac{1}{r\tilde{}\,-1}\right)}}$$

```
> simplify( " );
```

$$\lim_{t\to\infty} \mathrm{p}(\,t\,) = b\tilde{}^{\left(-\frac{1}{r\tilde{}\,-1}\right)}\, a\tilde{}^{\left(\frac{1}{r\tilde{}\,-1}\right)}$$

The dependence on $\frac{a}{b}$ confirms the observation that the solutions obtained by the improved Euler's method converge to powers of 3. The limiting population is readily seen to be independent of the initial population p_0. The fact that $r = 1$ is of special interest is also apparent.

While this completes the investigation of the selected questions, it need not be the end of the investigation. Slight modifications of the approach outlined above can be used to complete the analysis for all $0 < r < 1$.

A 3-d plot of the solution (with independent variables t and r) is a nice way to conclude this exercise.

```
> plot3d( subs( {a=3,b=1,p0=1}, SOLN ), t=0..5, r=1..4, view=0..10,
>         orientation=[130,80], axes=BOXED, style=WIREFRAME, shading=ZHUE,
>         title=`Figure 3` );
```



A particularly curious student, or reader, might want to see the plot for values of r on both sides of 1. Go ahead, try it!

Two Alternate Approaches to Numerical Solutions

Maple does provide several built-in numerical ODE solvers that could be used should you wish to avoid Maple programming. The share library contains an ODE package with implementations of Euler's method (`firsteuler`), improved Euler's method (`impeuler`) and the fourth-order Runge-Kutta method (`rungekuttahf`). Here is how these procedures could be used to replace `IEM`:

```
> with( share ): readshare(ODE,plots):
See ?share and ?share,contents for information about the share library
```

```
> a := 3: b := 1:
```

```
> r := 3/2:
> s1a := impeuler(f,[0,1],0.25,20);
```

$$s1a := \text{array}(0..20, [$$
$$(0) = [0, 1.]$$
$$(1) = [.25, 1.582860337]$$
$$(2) = [.50, 2.351441074]$$
$$(3) = [.75, 3.267497751]$$
$$(4) = [1.00, 4.253155546]$$
$$(5) = [1.25, 5.216751357]$$
$$(6) = [1.50, 6.083401955]$$
$$(7) = [1.75, 6.811626143]$$
$$(8) = [2.00, 7.392146417]$$
$$(9) = [2.25, 7.837089763]$$
$$(10) = [2.50, 8.168506666]$$
$$(11) = [2.75, 8.410361872]$$
$$(12) = [3.00, 8.584317467]$$
$$(13) = [3.25, 8.708165353]$$
$$(14) = [3.50, 8.795710426]$$
$$(15) = [3.75, 8.857285190]$$
$$(16) = [4.00, 8.900442869]$$
$$(17) = [4.25, 8.930618540]$$
$$(18) = [4.50, 8.951681525]$$
$$(19) = [4.75, 8.966366415]$$
$$(20) = [5.00, 8.976596163]$$
$$])$$

```
> r := 2: s2a := impeuler(f,[0,1],0.25,20):
```

```
> r := 3: s3a := impeuler(f,[0,1],0.25,20):
```

```
> plot( {makelist(s1a,1,2), makelist(s2a,1,2), makelist(s3a,1,2)},
>       title=`Figure 4 -- same as Figure 1` );
```

A second possibility is to use `dsolve` with the `type=numeric` option. See `?dsolve,numeric` for more information.

```
> h := 0.25: N := 20:
> Tvals := linalg[vector](  [ seq( h*i, i=0..N ) ] ):
```

```
> p0 := 1:
```

```
> r := 3/2: s1b := dsolve( {ode, ic}, p(t), type=numeric, value=Tvals );
```

$$s1b := \left[\begin{array}{c} [\, t\, \mathrm{p}(\, t\,)\,] \\ \left[\begin{array}{cc} 0 & 1. \\ .2500000000000000 & 1.596133111657981 \\ .5000000000000000 & 2.379700090173954 \\ .7500000000000000 & 3.308570369247907 \\ 1. & 4.302782541514969 \\ 1.250000000000000 & 5.270890496792378 \\ 1.500000000000000 & 6.139015076015999 \\ 1.750000000000000 & 6.866301507391285 \\ 2. & 7.443778244378324 \\ 2.250000000000000 & 7.883975322752338 \\ 2.500000000000000 & 8.209556163068363 \\ 2.750000000000000 & 8.445160919766465 \\ 3. & 8.613020003184459 \\ 3.250000000000000 & 8.731306497830154 \\ 3.500000000000000 & 8.814022097387028 \\ 3.750000000000000 & 8.871555066551442 \\ 4. & 8.911424104255532 \\ 4.250000000000000 & 8.938981702506480 \\ 4.500000000000000 & 8.957995927960795 \\ 4.750000000000000 & 8.971099398998826 \\ 5. & 8.980121951104255 \end{array} \right] \end{array} \right]$$

```
> r := 2: s2b := dsolve( {ode, ic}, p(t), type=numeric, value=Tvals ):
> r := 3: s3b := dsolve( {ode, ic}, p(t), type=numeric, value=Tvals ):
> plot( { s1b[2,1], s2b[2,1], s3b[2,1] },
>       title=`Figure 5 -- same as Figure 1` );
```

Concluding Remarks

- It should be noted that the direction field provide only strong evidence that solutions approach a limiting population when $r > 1$. A rigorous proof of this result is possible assuming a familiarity with qualitative analysis. Unfortunately, the syllabus for this course does not leave much time for the discussion of qualitative analysis.

- Suppose the solution is needed only for one set of parameters, say $a = 3$, $b = 1$, $r = 1.5$, and $p0 = 1$. Can `dsolve` solve this problem?

```
> a := 'a': b := 'b': r := 'r': p0 := 'p0':
> simplify( dsolve( subs( {a=3, b=1, r=3/2, p0=1}, { ode, ic } ), p(t) ) );
```

$$\mathrm{p}(\,t\,) = 9\,\frac{1}{(\,-1 + 4\,\mathrm{e}^{(\,-3/2\,t\,)}\,)^2}$$

While this solution looks plausible on first glance, note that the denominator has a zero at $t = \frac{2}{3}\ln 4 > 0$. It is easy to see that this solution does not agree with either the direction field in Figure 2a or the explicit solution found earlier.

```
> simplify( subs( {a=3, b=1, r=3/2, p0=1}, SOLN ) );
```

$$9\,\frac{1}{(\,1 + 2\,\mathrm{e}^{(\,-3/2\,t\,)}\,)^2}$$

The cause of this problem is almost certainly a branch cut error. While this is disconcerting, it does serve to remind the user of the dangers of treating any software package as a “black box”.

- The `IEM` procedure is easily adapted to higher-order numerical methods such as the fourth-order Runge-Kutta method. It is equally easy to extend this procedure for use with systems of first-order differential equations. This makes a nice student project.

- The careful reader will note the different syntax that has been used when manipulating a Maple *equation*. Three of the following commands are valid. Which ones?

```
> (A=B)+C;
> (A=B)+(C=C);
> (A=B)*C;
> (A=B)^C;
> exp(A=B);
> map( exp, A=B );
```

The lack of a uniform syntax for the manipulation of *equations* can be very confusing. Each of the above should be accepted as a valid Maple command.

Acknowledgement

The author would like to express appreciation to Daniel Dubois of Waterloo Maple Software Technical Support for encouraging the development of this paper. The comments of Matt Miller, University of South Carolina, and Mike Monagan, ETH-Zurich, and Jim Herod, Georgia Institute of Technology, have been instrumental in the final preparation of the paper.

References

[1] R. Kent Nagel and Edward B. Saff, *Fundamentals of Differential Equations*, 3rd ed., Addison-Wesley, 1993.